Rules for integrands of the form $(a + b \operatorname{ArcTan}[c x^n])^p$

1: $\int (a + b \operatorname{ArcTan}[c x^n])^p \, dx$ when $p \in \mathbb{Z}^+ \land (n == 1 \lor p == 1)$

Derivation: Integration by parts

Basis: $\partial_x (a + b \operatorname{ArcTan}[c x^n])^p == b\, c\, n\, p \frac{x^{n-1} (a+b \operatorname{ArcTan}[c x^n])^{p-1}}{1+c^2 x^{2n}}$

Rule: If $p \in \mathbb{Z}^+ \land (n == 1 \lor p == 1)$, then

$$\int (a + b \operatorname{ArcTan}[c x^n])^p \, dx \longrightarrow x (a + b \operatorname{ArcTan}[c x^n])^p - b\, c\, n\, p \int \frac{x^n (a + b \operatorname{ArcTan}[c x^n])^{p-1}}{1 + c^2 x^{2n}} \, dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_^n_.])^p_.,x_Symbol] :=
  x*(a+b*ArcTan[c*x^n])^p -
  b*c*n*p*Int[x^n*(a+b*ArcTan[c*x^n])^(p-1)/(1+c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,n},x] && IGtQ[p,0] && (EqQ[n,1] || EqQ[p,1])
```

```
Int[(a_.+b_.*ArcCot[c_.*x_^n_.])^p_.,x_Symbol] :=
  x*(a+b*ArcCot[c*x^n])^p +
  b*c*n*p*Int[x^n*(a+b*ArcCot[c*x^n])^(p-1)/(1+c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,n},x] && IGtQ[p,0] && (EqQ[n,1] || EqQ[p,1])
```

2. $\int (a + b \operatorname{ArcTan}[c\, x^n])^p \, dx$ when $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}$

1: $\int (a + b \operatorname{ArcTan}[c\, x^n])^p \, dx$ when $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\operatorname{ArcTan}[z] == \frac{i \operatorname{Log}[1 - i z]}{2} - \frac{i \operatorname{Log}[1 + i z]}{2}$

Basis: $\operatorname{ArcCot}[z] == \frac{i \operatorname{Log}[1 - i z^{-1}]}{2} - \frac{i \operatorname{Log}[1 + i z^{-1}]}{2}$

Rule: If $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}^+$, then

$$\int (a + b \operatorname{ArcTan}[c\, x^n])^p \, dx \longrightarrow \int \operatorname{ExpandIntegrand}\left[\left(a + \frac{i\, b \operatorname{Log}[1 - i\, c\, x^n]}{2} - \frac{i\, b \operatorname{Log}[1 + i\, c\, x^n]}{2}\right)^p, x\right] dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+(I*b*Log[1-I*c*x^n])/2-(I*b*Log[1+I*c*x^n])/2)^p,x],x] /;
FreeQ[{a,b,c},x] && IGtQ[p,1] && IGtQ[n,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+(I*b*Log[1-I*x^(-n)/c])/2-(I*b*Log[1+I*x^(-n)/c])/2)^p,x],x] /;
FreeQ[{a,b,c},x] && IGtQ[p,1] && IGtQ[n,0]
```

2: $\int (a + b \operatorname{ArcTan}[c\, x^n])^p \, dx$ when $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: $\operatorname{ArcTan}[z] == \operatorname{ArcCot}\left[\frac{1}{z}\right]$

Rule: If $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}^-$, then

$$\int (a + b\,\text{ArcTan}[c\,x^n])^p\,dx \longrightarrow \int \left(a + b\,\text{ArcCot}\left[\frac{x^{-n}}{c}\right]\right)^p\,dx$$

–

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_^n_])^p_,x_Symbol] :=
  Int[(a+b*ArcCot[x^(-n)/c])^p,x] /;
FreeQ[{a,b,c},x] && IGtQ[p,1] && ILtQ[n,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_^n_])^p_,x_Symbol] :=
  Int[(a+b*ArcTan[x^(-n)/c])^p,x] /;
FreeQ[{a,b,c},x] && IGtQ[p,1] && ILtQ[n,0]
```

3: $\int (a + b\,\text{ArcTan}[c\,x^n])^p\,dx$ when $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[x] == k\,\text{Subst}\left[x^{k-1}\,F\left[x^k\right],\,x,\,x^{1/k}\right]\,\partial_x x^{1/k}$

–

Rule: If $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{F}$, let $k \to \text{Denominator}[n]$, then

$$\int (a + b\,\text{ArcTan}[c\,x^n])^p\,dx \longrightarrow k\,\text{Subst}\left[\int x^{k-1}\,\left(a + b\,\text{ArcTan}\left[c\,x^{k\,n}\right]\right)^p\,dx,\,x,\,x^{1/k}\right]$$

–

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_^n_])^p_,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*ArcTan[c*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c},x] && IGtQ[p,1] && FractionQ[n]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_^n_])^p_,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*ArcCot[c*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c},x] && IGtQ[p,1] && FractionQ[n]
```

U: $\int (a + b \operatorname{ArcTan}[c x^n])^p \, dx$

Rule:

$$\int (a + b \operatorname{ArcTan}[c x^n])^p \, dx \longrightarrow \int (a + b \operatorname{ArcTan}[c x^n])^p \, dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_^n_.])^p_,x_Symbol] :=
  Unintegrable[(a+b*ArcTan[c*x^n])^p,x] /;
FreeQ[{a,b,c,n,p},x]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_^n_.])^p_,x_Symbol] :=
  Unintegrable[(a+b*ArcCot[c*x^n])^p,x] /;
FreeQ[{a,b,c,n,p},x]
```